UCID- 18513

A JOINT GAUSSIAN PROBABILITY PLOT PROGRAM

S. G. Azevedo
D. T. Gavel

January 1980

Lawrence Livermore Laboratory

Work performed under the auspices of the U.S. Department of Energy by the Lawrence Livermore Laboratory under Contract W-7405-Eng-48.

ABSTRACT

The analysis of joint-Gaussian distributions of two variables can be aided greatly by meaningful graphical techniques. The program described by this report performs such a task. Two elliptical equi-probability contour plots of the probability density function are produced; one using statistics from the raw data and the other using normalized (zero mean, unit variance) variables. Several separate data sets of the same variables may be plotted, for example, to compare estimators. The techniques discussed here for interpreting the plots enable the user to gain insight into the statistical information being displayed.

Table of Contents

List of Figures

List of Tables

I. INTRODUCTION

Analysis of errors in measurements, estimates, parameters, etc., is frequently encountered in engineering applications. By virtue of the Central Limit Theorem [1], many of these errors can reasonably be assumed to have normal (Gaussian) distributions. In order to ease the burden of numerical error analysis, graphical methods of displaying the important statistical information have been developed.[2] This report describes such a method for the two-dimensional Gaussian vector case.

When confronted with a single Gaussian-distributed random variable, it is not difficult to draw or visualize the probability density function (pdf) given the mean and standard deviation. This allows easy computation of confidence intervals or probability intervals as is done often in sample statistics. However, with bivariate Gaussian random variables, the pdf becomes projected in three-space and the statistical properties are much less intuitively obvious. We now become concerned with covariance and correlation [1], in addition to the properties of the individual random variables.

The computer program described in this report, ECP2D, presents a meaningful way of displaying these joint distributions. This is done by plotting contours of constant probability density which have certain user-specified probability of containment within the contour. In this way, the important information from the three dimensional probability density surface is projected onto a planar coordinate system as is done with topographical maps. In our case, we are concerned only with Gaussian distributions, so these contours are all elliptical in shape.[3] The projection is shown in Figure 1.

If many separate experiments or trials each produce statistics describing the joint pdf of two random variables, the contours of each trial may be displayed on one plot for comparison. This comparison provides the motivation for development of the ECP2D program. The ellipses produced can be a means of ascertaining the performance of one parameter estimator over another. However, it is sufficiently general to be applicable to any bivariate Gaussian random vectors.

Automatic scaling of the plot, and differences in units of the two axes can cause distortion in the size and orientation of the contours, so an alternative graphing method showing normalized data was also developed. This provides much more information about the correlation of the two random variables.



FIGURE 1: ELLIPTICAL CONTOURS OF THE JOINT DENSITY FUNCTION IN (x_1, x_2)

In this report, we discuss:

- Contour Plotting Theory
- Program Usage
- Interpretation of the Output

II. THEORY

In this section, we develop the theory behind plotting an ellipse of constant probability density. The fundamental problem can be stated as:

> Given the mean and the covariance matrix of a joint Gaussian random vector in two-space, find an equi-probability elliptical contour such that the probability of containment within the ellipse is some value, P_A.

In order to solve this problem we:

1) Show that a constant probability contour for a Gaussian pdf is an ellipse;
2) Transform the ellipse to the origin;
3) Normalize the axes (circle);
4) Reconstruct the original ($\underline{x}$-plane) ellipse for the calculated value of P_A; and
5) Calculate size of the ellipse.

After plotting this ellipse, a second plot is also generated for displaying correlation between the two random variables.

II-1. Ellipse Equation

With the assumption of Gaussian random variables (call them x_1 and x_2), the joint probability density function is

$$p(x_1,x_2) = \frac{1}{2\pi|\underline{C}|^{1/2}} \exp\left[-1/2(\underline{x}-\underline{\mu}_x)^T \underline{C}^{-1}(\underline{x}-\underline{\mu}_x)\right] \qquad (1)$$

where $\underline{x} = \begin{bmatrix} x_1 \\ x_2 \end{bmatrix}$ = Gaussian random vector

$\underline{\mu}_x = \begin{bmatrix} \mu_1 \\ \mu_2 \end{bmatrix} = \begin{bmatrix} E(x_1) \\ E(x_2) \end{bmatrix}$ = mean value of vector $\underline{x}$

$\underline{C} = \begin{bmatrix} c_{11} & c_{12} \\ c_{21} & c_{22} \end{bmatrix}$ = covariance matrix

that is $c_{ij} = E\left[(x_i-\mu_i)\ (x_j-\mu_j)\right]$ for $i,j = 1,2$

so $c_{12} = c_{21}$.

Making $p(x_1,x_2)$ a constant and solving for the quadratic form in (1) we obtain:

$$(\underline{x}-\underline{\mu}_x)^T \; \underline{C}^{-1}(\underline{x}-\underline{\mu}_x) = b^2 \qquad (2)$$

Expanding (2) we have the equation

$$\alpha(x_1-\mu_1)^2 + \beta(x_2-\mu_2)^2 + \gamma(x_1-\mu_1)\,(x_2-\mu_2) = b^2$$

where $\alpha = \dfrac{c_{22}}{c_{11}c_{22} - c_{12}^2}$

$$\beta = \frac{c_{11}}{c_{11}c_{22} - c_{12}^2}$$

$$\gamma = \frac{2c_{12}}{c_{11}c_{22} - c_{12}^2}$$

and b = scalar constant.

The matrix $\underline{C}$ is constrained to be positive definite [4] because it is a covariance matrix. This constraint causes (2) to be the equation for an ellipse in the $\underline{x}$ plane. The constant b is related to the total probability distributed within the ellipse, P_A, (to be shown in Section II-5). We will assume that b is an arbitrary constant for now and explain the procedure for locating and plotting the ellipse defined by equation (2).

II-2. Rotation and Translation of the Ellipse

To simplify the computation of the ellipse equations, we can perform a similarity transformation upon the $\underline{C}$ matrix. We constrain the transformed matrix to be diagonal which merely rotates the coordinate axes to align with the major and minor axes of the ellipse. Then subtracting the mean vector from $\underline{x}$ translates the center of the ellipse to the origin (see Figure 2).

The transformation we wish to make is:

$$\underline{y} = \underline{S}^{-1} \, (\underline{x}-\underline{\mu}_x) \qquad (3)$$

where $\underline{y} = \begin{bmatrix} y_1 \\ y_2 \end{bmatrix}$ = the transformed variable

and $\underline{S}$ is the transformation matrix.

Then from equations (2) and (3),

$$(\underline{S}\underline{y})^T \, \underline{C}^{-1} \, (\underline{S}\underline{y}) = b^2$$

or

$$\underline{y}^T \underline{D}^{-1} \underline{y} = b^2 \qquad (4)$$



$$(\underline{x}-\underline{\mu}_x)\,\underline{C}^{-1}\,(\underline{x}-\underline{\mu}_x) = b^2$$

$$\left\{\alpha(x_1-\mu_1)^2+\beta(x_2-\mu_2)^2-\gamma(x_1-\mu_1)(x_2-\mu_2)=b^2\right\}$$

Original Ellipse in $\underline{x}$-plane

$$\underline{y} = \underline{S}^{-1}(\underline{x}-\underline{\mu}_x),\quad \underline{S}=\begin{bmatrix}\cos\theta & -\sin\theta\\ \sin\theta & \cos\theta\end{bmatrix}$$



$$\frac{y_1^2}{(d_1 b)^2}+\frac{y_2^2}{(d_2 b)^2}=1$$

Translated and Rotated to the $\underline{y}$-plane

$$\underline{z}=\underline{T}^{-1}\,\underline{y},\quad \underline{T}=\begin{bmatrix}d_1 & 0\\ 0 & d_2\end{bmatrix}$$



$$z_1^2+z_2^2=b^2$$

Normalized to the $\underline{z}$-plane (circle)

FIGURE 2: TRANSFORMATIONS OF THE ELLIPSE

where
$$\underline{D} = \begin{bmatrix} d_1^2 & 0 \\ 0 & d_2^2 \end{bmatrix} = \underline{S}^{-1}\ \underline{C}\ \underline{S}^{-T}$$

Notice that d_1^2 and d_2^2 are the variances with respect to the $\underline{y}$-coordinate system. Equation (4) can be written alternatively as:

$$\frac{y_1^2}{d_1^2} + \frac{y_2^2}{d_2^2} = b^2$$

so that (bd_1) and (bd_2) are the major and minor axes of the translated-rotated ellipse (see Figure 2).

In order for the joint probability density function to remain the same throughout this transformation, $\underline{D}$ and $\underline{C}$ must be similar matrices [3] ($\underline{C} = \underline{SDS}^{-1}$). Under a similarity transformation, the determinant and eigenvalues are unchanged. Since $\underline{C}$ is a symmetric matrix, this similarity transformation specializes to an orthogonal transformation where $\underline{S}^T = \underline{S}^{-1}$. An $\underline{S}$ matrix which satisfies these conditions is:

$$\underline{S} = \begin{bmatrix} \cos\theta & -\sin\theta \\ \sin\theta & \cos\theta \end{bmatrix} \qquad (5)$$

where θ is the angle of rotation of the $\underline{y}$-axes from the $\underline{x}$-coordinate plane. Now, from this transformation we can derive that [3] (see Appendix B for details)

$$\left.\begin{matrix} d_1^2 \\ \\ d_2^2 \end{matrix}\right\} = \frac{c_{11} + c_{22} \pm \sqrt{(c_{11} - c_{22})^2 + 4(c_{12})^2}}{2} \qquad (6)$$

$$\theta = 1/2 \quad \tan^{-1}\left[\frac{2c_{12}}{c_{11} - c_{22}}\right] \qquad (7)$$

II-3. Normalization of the Ellipse

With a second transformation, we can easily normalize to a circle of radius b (see Figure 2). The reasons for doing this will become apparent later. We set

$$\underline{z} = \underline{T}^{-1}\underline{y} \qquad (8)$$

where $\underline{z} = \begin{bmatrix} z_1 \\ z_2 \end{bmatrix}$ = the transformed vector

and $\underline{T} = \begin{bmatrix} d_1 & 0 \\ 0 & d_2 \end{bmatrix}$ (9)

Then from equations (4) and (8),

$$(\underline{Tz})^T \underline{D}^{-1} (\underline{Tz}) = b^2$$

$$\underline{z}^T(\underline{T}^T \underline{D}^{-1}\underline{T})\underline{z} = b^2$$

But by the definitions, $\underline{T}^T\underline{D}^{-1}\underline{T} = \underline{I}$ (identity matrix) so we obtain the circle equation (Figure 2)

$$\underline{z}^T\underline{z} = b^2 \qquad (10)$$

or $\quad z_1^2 + z_2^2 = b^2$

II-4. Reconstruct Original Ellipse

Now, given the points z_1 and z_2 which satisfy this equation, we can easily reconstruct the $\underline{x}$ vector using equations (2) and (8) which give:

$$\underline{x} = \underline{STz} + \underline{\mu}_x \tag{11}$$

This now illustrates the simple way in which ECP2D calculates the plot points for an ellipse. First, an array of coordinate pairs representing the solution to equation (10) is established. Then, by determining d_1, d_2, and θ, the points of the unit circle can be projected back to the $\underline{x}$-plane (Figure 2) with the transformation given in equation (11). We must be careful however that θ is in the correct quadrant for valid reconstruction.

II-5. Calculate Size of Ellipse (determine b)

Our problem of calculating the integrated probability within the ellipse, $(\underline{x}-\underline{\mu}_x)^T \, \underline{C}^{-1}(\underline{x}-\underline{\mu}_x) = b^2$, has now been simplified to computing the probability within the circle $\underline{z}^T\underline{z} = b^2$. To find this probability, P_A (and the corresponding value of b), we integrate the joint probability density function over the area of the circle. This can be shown by starting from equation (1) and integrating in the $\underline{x}$ space:

$$p_A = \iint_A \frac{1}{2\pi|\underline{C}|^{1/2}} \exp\left[-1/2\ (\underline{x}-\underline{\mu}_x)^T\ \underline{C}^{-1}(x-\underline{\mu}_x)\right]\ dx_1dx_2 \qquad (12)$$

where A is the interior area of the ellipse defined by

$$(\underline{x}-\underline{\mu}_x)^T\ \underline{C}^{-1}(\underline{x}-\underline{\mu}_x) = b^2$$

Now, from the definitions of $\underline{T}$ and $\underline{S}$ by the transformations described earlier,

$$\underline{C} = \underline{ST}\underline{T}^T\underline{S}^T$$

$$\text{or} \quad \underline{C}^{-1} = \underline{S}^{-T}\underline{T}^{-T}\underline{T}^{-1}\underline{S}^{-1} \qquad (13)$$

But, also from equation (11),

$$(\underline{x}-\underline{\mu}_x) = \underline{STz} \qquad (14)$$

$$\text{and} \quad (\underline{x}-\underline{\mu}_x)^T = \underline{z}^T\underline{T}^T\underline{S}^T \qquad (15)$$

So, substituting (13), (14), and (15) into (12) we get

$$p_A = \iint_A \frac{1}{2\pi|\underline{C}|^{1/2}} \exp\left[-1/2\ \underline{z}^T\underline{z}\right] dx_1dx_2 \qquad (16)$$

The next step is to change the variables of integration so that we integrate over the z-space. The formula for this is [6]:

$$p_A = \iint_A \frac{1}{2\pi |\underline{C}|^{1/2}} \exp\left[-1/2\ \underline{z}^T\underline{z}\right] \left|\frac{\partial x}{\partial z}\right| \partial z_1 \partial z_2 \qquad (17)$$

where $\frac{\partial x}{\partial z} \triangleq \begin{bmatrix} \frac{\partial x_1}{\partial z_1} & \frac{\partial x_1}{\partial z_2} \\ \frac{\partial x_2}{\partial z_1} & \frac{\partial x_2}{\partial z_2} \end{bmatrix}$.

It is easy to show, using Equation (14), that

$$\left|\frac{\partial x}{\partial z}\right| = |\underline{C}|^{1/2}$$

so,

$$p_A = \iint_A \frac{1}{2\pi} \exp\left[-1/2\ \underline{z}^T\underline{z}\right] dz_1 dz_2 \qquad (18)$$

The integration area, A, is the area inside the circle, $\underline{z}^T\underline{z} = b^2$. For easy integration, we can convert to polar coordinates:

$$P_A = \int_{r=0}^{b} \int_{\varphi=-\pi}^{\pi} \frac{1}{2\pi} \exp\left(-1/2 r^2\right) r d\varphi dr$$

$$= \int_0^b r e^{-\frac{1}{2} r^2} dr \qquad (19)$$

where r = radial distance = $\sqrt{z_1^2 + z_2^2}$

φ = angle from z_1-axis = $\tan^{-1}\left(\frac{z_2}{z_1}\right)$

Performing the integration we arrive at

$$P_A = 1 - e^{-\frac{1}{2} b^2}$$

then solving for b^2

$$b^2 = -2\ln (1-P_A) \tag{20}$$

If we wish, for example, to plot the ellipse within which 95% of the $\underline{x}$ values can be expected to lie, we use equation (2) with b^2 set equal to

$$b^2 = -2\ln (1-.95) = 5.99$$

When b = 1,2, or 3, we encounter the often used one-, two- or three-sigma probability contours [3,5] (see Table I).

Table I: b-Sigma Probability Distribution Values

b	P_A
1	.394
2	.865
3	.989

II-6. Correlation Plot

In an effort to obtain more visual information from the data, we present a second contour plot we call a correlation plot. For this plot, another transformation is performed to a coordinate system where the variance in both directions is normalized to 1. We merely show here how the transformation is implemented, and present later the interpretation of the graph.

The elliptical plot produced in the previous sections graphically illustrates a contour of particular likelihood for the vector $\underline{x}$. But what if the units or the orders of magnitude of the two varibles, x_1 and x_2, are vastly different? We may still wish to see the first plot of the original two-space ellipse, but distortion caused by scaling may give misleading information with resoect to correlation or dependence of the two random variables.

For that reason, a second plot is created which gives normalized information for a better display of the comparisons between x_1 and x_2. This is performed by the following transformation.

$$\underline{x}^* = \underline{N}\,(\underline{x}-\underline{\mu}_x) \qquad (21)$$

where $\underline{x}^* = \begin{bmatrix} x_1^* \\ x_2^* \end{bmatrix}$ = the normalized vector

and $\underline{N}$ is the normalization matrix.

Then from equation (2)

$$(\underline{N}^{-1}\underline{x}^*)^T\,\underline{C}^{-1}(\underline{N}^{-1}\underline{x}^*) = b^2$$

or

$$\underline{x}^{*T}\underline{G}^{-1}\underline{x}^* = b^2$$

or $(x_1^*)^2 + (x_2^*)^2 - 2\rho x_1^* x_2^* = b^2(1-\rho^2)$ $\qquad (22)$

where

$$\underline{G} = \begin{bmatrix} 1 & \rho \\ \rho & 1 \end{bmatrix} = \underline{NCN}^T = \text{Correlation matrix}$$

$$\rho = \frac{c_{12}}{\sqrt{c_{11}c_{22}}} = \text{Correlation coefficient}$$

so by algebraic manipulation

$$\underline{N} = \begin{bmatrix} \frac{1}{\sqrt{c_{11}}} & 0 \\ 0 & \frac{1}{\sqrt{c_{22}}} \end{bmatrix}$$

This gives us a new Gaussian random vector, $\underline{x}^*$, (see Figure 3), with zero mean and unit variance. The plot of the contour in the $\underline{x}^*$-space gives us an indication of the correlation of the original vector, $\underline{x}$. An explanation on how to interpret this data is given in the next section.



FIGURE 3: TRANSFORMATION TO CORRELATION PLOT

III. PROGRAM EXECUTION

This section explains the usage of ECP2D and how to interpret the output. A general flow diagram of the program is shown in Figure 4. Also presented is a sample application involving estimation schemes.

III-1. User Options

The program has been designed for ease and flexibility of use. Several possible input options are available to the user for producing the output desired.

For a bivariate normally distributed random vector $\underline{x}$ with mean $\underline{\mu}_x$ and covariance matrix $\underline{C}$, let $\underline{x}(1),\underline{x}(2),\ldots,\underline{x}(N)$ be a set of data. Many such data sets may be processed and plotted for either of two cases:

1) The data $\underline{x}(1),\underline{x}(2),\ldots,\underline{x}(N)$ is preprocessed (i.e., $\underline{\mu}_x$ and $\underline{C}$ are known, or estimates of $\underline{\mu}_x$ and $\underline{C}$ are available) and then only $\underline{\mu}_x$ and $\underline{C}$ are inputted to the program for each data set. The $\underline{C}$ matrix <u>must</u> be positive definite.

2) The raw data $\underline{x}(1),\underline{x}(2),\ldots,\underline{x}(N)$ is inputted for each data set.



FIGURE 4: GENERAL FLOW CHART OF ECP2D

Up to ten different ellipses, corresponding to ten probability values, may be plotted <u>for each</u> data set on the same graph. In addition, any constant coordinate pair (labelled x_{1true}, x_{2true}) can be plotted for comparison.

III-2. Input File Format

The input file for ECP2D must be present on the disc at the start of execution and must be named ECPIN. This file allows the user to specify certain plot options as well as the problem parameters. Table II gives a description of each field in the input file. Table III lists the input fields and their corresponding formats. Notice that lines four and five are repeated for each of the data sets to be input.

III-3. Sample Input for ECP2D

An application of this plotting program is to compare the performance of several estimation schemes for identifying some constant parameters (θ_1,θ_2) of an arbitrary mathematical system. For this problem:

1) Each data set corresponds to N trials of an estimation scheme.

Table II: Input File Field Description

<u>Line 1</u> General Parameters for Plot

M	-	Number of data sets to be input
L	-	Number of contours to draw for each data set
LBLX	-	Label for the x_1-axis of the plot
LBLY	-	Label for the x_2-axis of the plot

<u>Line 2</u> True Value of the Vector x

ITFLAG	-	Do you know the true value of <u>both</u> x_1 and x_2? (Ø = NO, 1 = YES)
X1T	-	True value of x_1 (if known)
X2T	-	True value of x_2 (if known)

<u>Line 3</u> Probability Values for the Plot

PR(1), ..., PR(L) - Probability values for each contour (must be between 0 and 1)

REPEAT THE FOLLOWING LINES FOR EACH DATA SET (M times)

<u>Line 4</u> Data Set Input Parameters

MP	-	Is input (for this data set) raw data or preprocessed? (Ø = Raw, 1 = Preprocessed)
N	-	Number of data points (if MP = 0)

<u>Line 5</u> Data Set Input

<u>If MP = 0:</u>

X1(1),X2(1), ..., X1(N), X2(N) - Raw Data Points; $\underline{x}(1),\ldots,\underline{x}(N)$

<u>If MP = 1:</u>

XE1,XE2 - Mean Values of x_1 and x_2

P11,P22 - Variance of x_1 and x_2

P12 - Covariance of x_1 and x_2

Table III: Format of Input File

Line	Variable Name		Format
1	M,L,LBLX,LBLY		(2I5,2A10)
2	ITFLAG,X1T,X2T		(I2,2E16.7)
3	PR(1),....,PR(L)		(10F6.4)
4	MP,N		(I2,I5)
5	If MP = 0	X1(1),X2(1),...X1(N),X2(N)	(2E16.7)
	If MP = 1	XE1,XE2,P11,P22,P12	(5E16.7)

2) $\underline{x}_i(1),\ldots,\underline{x}_i(N)$ are estimates of $\underline{\theta} = \left[\theta_1\theta_2\right]^T$ for each set i, and the sample mean and covariance are used to approximate the $(\underline{\mu}_x)_i$ and $\underline{C}_i$.

3) The true (known) value of $\underline{\theta}$ is plotted as $\underline{x}_{true}$.

The input file for this application, with two estimators (data sets), is shown in Table IV. The data shown in lines 4-14 and lines 15-16 can be thought of as estimated values of parameters θ_1 and θ_2 from two different estimation schemes. Notice that the first data set is in raw form with N=10 and the second data set is given in "pre-processed" form. A discussion of the output is given in the next section.

III-4. Sample Problem

The standard output obtained from ECP2D is shown in Figures 5 through 7. Figure 5 gives pertinent self-explanatory numeric information with regard to the plots. The probability values used are the one-sigma, two-sigma, and three-sigma values for a two-dimensional Gaussian random

Table IV: Example Problem Input File

Line #

```
2,3,THETA 1   THETA 2
1,1.E-08,5.E-07,
.394,.865,.989,
0,10,
1.003760E-08,4.8368E-07,
1.004390E-08,4.8295E-07,
1.005415E-08,4.8183E-07,
1.005429E-08,4.8174E-07,
1.005456E-08,4.8181E-07,
1.005590E-08,4.8172E-07,
1.005790E-08,4.8165E-07,
1.005174E-08,4.8209E-07,
1.004282E-08,4.8309E-07,
1.005730E-08,4.8156E-07,
1,,
9.99E-09,5.04E-07,7.E-23,8.E-19,5.E-21,
```

vector. The angle of the major axis of the ellipse from the x_1 axis (between -90° and $+90^\circ$) is calculated and displayed along with the diagonalized variances (d_1^2 and d_2^2; the eigenvalues of $\underline{C}$). Also, the correlation coefficient (ρ) is shown; if $\rho = 1$, we have maximum correlation between x_1 and x_2, if $\rho = -1$ we have maximum negative correlation, and if $\rho = 0$ we have no correlation between the two variables.

In Figure 6, the plots of the two data sets are displayed. The size of the two contours gives an indication of the relative precision of each estimator; i.e., the curves for data set A are smaller than those for data set B, so estimator A is more precise in its estimate. Accuracy of the estimators, on the other hand, can be observed by the relative distance of mean value points from the true value (the large black dot). This is called the bias of the estimator and it indicates low accuracy when the bias is large. Notice in Figure 6 that although estimator A produces more precise estimates than B, there is also a larger bias in the A estimates indicating less accuracy.

Note, also, that the apparent angle of rotation of the ellipses does not correspond to that reported on Figure 5. This is due to the distortion caused by scaling since x_1 and x_2 have vastly different units and values.

```
                    TWO-SPACE PROBABILITY PLOT

                       TRUE VALUE OF X1 :    1.000E-08
                       TRUE VALUE OF X2 :    5.000E-07

                  PROBABILITY VALUES OF CONTOURS
0.3940  0.8650  0.9890


DATA SET NO.  1
                      SAMPLE MEAN OF X1 :    1.005E-08
                      SAMPLE MEAN OF X2 :    4.822E-07
                  SAMPLE VARIANCE OF X1 :    4.913E-23
                  SAMPLE VARIANCE OF X2 :    5.554E-19
                      SAMPLE COVARIANCE :   -5.196E-21
                CORRELATION COEFFICIENT :   -9.948E-01
  ANGLE OF MAJOR AXIS FROM X1 (DEGREES) :  -89.4640
                 DIAGONALIZED VARIANCES :    5.555E-19   5.135E-25

DATA SET NO.  2
                      SAMPLE MEAN OF X1 :    9.990E-09
                      SAMPLE MEAN OF X2 :    5.040E-07
                  SAMPLE VARIANCE OF X1 :    7.000E-23
                  SAMPLE VARIANCE OF X2 :    8.000E-19
                      SAMPLE COVARIANCE :    5.000E-21
                CORRELATION COEFFICIENT :    9.572E-01
  ANGLE OF MAJOR AXIS FROM X1 (DEGREES) :   89.6419
                 DIAGONALIZED VARIANCES :    8.000E-19   3.875E-23
```

Figure 5: Typical Output Information Display



Figure 6: Display of Contours for Two Data Sets



Figure 7: Correlation Plots of the Data Sets

Therefore, we can obtain general relative information about the properties of the two data sets, and directly read the means and standard deviations (max x_1 value on the one-sigma plot minus μ_1, is σ_1), but we cannot interpret the correlation of the two variables from this plot.

For that purpose, we turn to Figure 7 which displays the normalized results. Here we plot

$$\underline{x}^* = \underline{N}(\underline{x}-\underline{\mu}_x)$$

or $$x_1^* = \frac{x_1-\mu_1}{\sigma_1} \qquad \text{and} \qquad x_2^* = \frac{x_2-\mu_2}{\sigma_2}$$

where σ_1 and σ_2 are the standard deviations of x_1 and x_2 respectively. Now the two variables are dimensionless so the relationship between them is more apparent. In fact, only the correlation coefficient (ρ) is available from the graph. Figure 8 shows how different correlation coefficients affect the orientation of the graph.

Thus, we see that the θ_1 parameter and θ_2 parameter are highly (but negatively) correlated in data set A (of Figure 7), but not as highly (and positively) correlated in data set B. So with one estimator, the estimates are very highly dependent on one another, but less so with the other estimator.



PERFECT POSITIVE CORRELATION
($\rho = 1$)

PARTIAL POSITIVE CORRELATION
($0 < \rho < 1$)

NO CORRELATION; UNCORRELATED
($\rho = 0$)

PARTIAL NEGATIVE CORRELATION
($-1 < \rho < 0$)

PERFECT NEGATIVE CORRELATION
($\rho = -1$)

FIGURE 8 : DETERMINING CORRELATION OF TWO RANDOM VARIABLES FROM THE NORMALIZED CONTOUR PLOT

IV. SUMMARY

We have shown how to display the statistical information contained in a Gaussian random vector in two-space. The computer program ECP2D was developed as a data display/interpretation tool for use specifically in comparing parameter estimates [7].

ACKNOWLEDGMENTS

Many thanks to the motivating efforts of J. V. Candy and the technical statistical help of R. W. Mensing. We would also like to thank Ms. Gail Simpson for typing the manuscript.

REFERENCES

1. Bendat, J. S. and Piersol, A. G., Random Data: Analysis and Measurement Procedures, New York, John Wiley & Sons, Inc., 1971.

2. Davis, J. C., Statistics and Data Analysis in Geology, New York, John Wiley & Sons, Inc., 1973.

3. Bryson, A. E., Jr. and Ho, Y., Applied Optimal Control: Optimization, Estimation, and Control, Washington, D. C., Hemisphere Publishing Corp., 1975.

4. Papoulis, A., Probability, Random Variables, and Stochastic Processes, New York, McGraw-Hill Book Company, 1965.

5. Gura, I. A. and Gersten, R. H., "Interpretation of n-Dimensional Covariance Matrices," AIAA Journal, Vol. 9, No. 4, pp. 740-742, April, 1971.

6. Hildebrand, F. B., Advanced Calculus for Applications, Englewood Cliffs, New Jersey, Prentice-Hall, Inc., 1962.

7. Azevedo, S. G., Candy, J. V. and Pimentel, K. D., "Parameter Identification in Hydrology ," UCID-Report, Circa 1979.

APPENDIX A

Source Code Listing for ECP2D

```
$CHAT (TV80LIB,ORDERLIB,STACKLIB,EE, ^) %ME X%ME B%ME L%ME N93   S S 80

      PROGRAM ECP2D(ECPIN,TAPE5=ECPIN)

C*******************************************************************
C*
C*    DESCRIPTION
C*       THIS PROGRAM PLOTS ELLIPSES OF CONSTANT LIKELIHOOD
C*    FOR TWO GAUSSIAN RANDOM VARIABLES.  A WRITE-UP ON THE
C*    THEORY OF OPERATION IS AVAILABLE IN A SEPARATE REPORT.
C*
C*    AUTHOR -- STEVE AZEVEDO       L-156     X2-8538
C*
C*    AVAILABILITY -- CDC 7600
C*
C*    EXECUTE LINE --    XECP2D / T V
C*
C*******************************************************************
C
C.....MN   - MAX NUMBER OF MEASUREMENTS
C.....ML   - MAX NUMBER OF CONTOURS TO PLOT
C.....MM   - MAX NUMBER OF DATA SETS TO PLOT
C.....NP   - NUMBER OF POINTS TO PLOT ON EACH ELLIPSE
      PARAMETER (MN=100,ML=10,MM=10)
      PARAMETER (NP=100)
      PARAMETER (PI=3.1415926535898)

      DIMENSION X1(MN,MM),X2(MN,MM)
      DIMENSION X(NP,ML,MM),Y(NP,ML,MM)
      DIMENSION PR(ML)
      DIMENSION XE1(MM),XE2(MM),S1(MM),S2(MM)
      DIMENSION WX(NP),WY(NP)
C
C***********************************************************
C*     PROGRAM INITIALIZATION
C***********************************************************
C
      CALL CHANGE("+XECP2D")
C
C.....INITIALIZE PLOTTING FOR FR80 105MM FICHE (BUT KEEP FIRST)
C
      CALL FR80ID(9HFILM-ONLY,0,1,0)
      CALL KEEP80(10H     ECPS0,3)
      CALL DDERS(-1)
      CALL SETCH (7.,42.,1,0,1,0)
      WRITE(100,"(20X,""TWO-SPACE PROBABILITY PLOT"")")
C
C.....COMPUTE POINTS OF A CIRCLE (RADIUS=1)
C
      DLTA = 2.*PI/(NP-1)
      ALPHA = 0.
      DO DL02 I=1,NP
      WX(I) = COS(ALPHA)
      WY(I) = SIN(ALPHA)
      ALPHA = ALPHA + DLTA
DL02  CONTINUE
C
C**********************************************************
C*     READ THE INPUT FILE (ECPIN)
C**********************************************************
```

```
C
C.....LINE 1. READ NUMBER OF DATA SETS AND CONTOURS
C           M - NUMBER OF DATA SETS
C           L - NUMBER OF CONTOURS
C        LBLX - LABEL FOR X-AXIS
C        LBLY - LABEL FOR Y-AXIS
        READ(5,"(2I5,2A10)") M,L,LBLX,LBLY
        IF (M.GT.MM)  THEN
              WRITE(59,"(1X,/,"" ERROR--TOO MANY DATA SETS; M = "",I5)") M
              CALL EXIT(1)
           ENDIF
        IF (L.GT.ML)  THEN
              WRITE(59,"(1X,/,"" ERROR--TOO MANY CONTOURS; L = "",I5)") L
              CALL EXIT(1)
           ENDIF
C
C.....LINE 2. READ TRUE VALUES OF X1 AND X2
C         ITFLAG = 1 IF TRUE VALUE IS KNOWN
C                  0 IF NOT
C         X1T    - TRUE VALUE OF X1
C         X2T    - TRUE VALUE OF X2
        READ(5,"(I2,2E16.7)") ITFLAG,X1T,X2T
        IF (ITFLAG.NE.0)  THEN
              WRITE(100,"(/,23X,""TRUE VALUE OF X1 : "",E12.3)") X1T
              WRITE(100,"(23X,""TRUE VALUE OF X2 : "",E12.3)") X2T
           ENDIF
C
C.....LINE 3. READ CONTOUR PROBS. (BETWEEN 0. AND 1.); SAME FOR EACH DATA SET
        READ(5,"(10F6.4)") (PR(I),I=1,L)
        WRITE(100,"(/,18X,""PROBABILITY VALUES OF CONTOURS"")")
        WRITE(100,"(10(F6.4,2X))") (PR(I),I=1,L)
C
        DO DL10 K=1,M
C
C.....NEXT LINES.  READ DATA SETS
C         MP   = 0 IF RAW DATA INPUT
C                1 IF ONLY X-MEAN AND COV. MATRIX INPUT
C         N    - NO. OF SAMPLE POINTS (IF MP=0)
        READ(5,"(I2,I5)") MP,N
        IF (MP.EQ.1)  THEN
              READ(5,"(5E16.7)") XE1(K),XE2(K),C11,C22,C12
              N=0
              IF ((C11*C22).LT.(C12*C12)) THEN
                    WRITE(59,"(/"" ERROR--COV MATRIX NOT POS DEF; SET #"",I5)") K
                    CALL EXIT(1)
                 ENDIF
           ELSE
              IF (N.LT.2) THEN
                    WRITE(59,"(/"" ERROR--NOT ENOUGH DATA; SET #"",I5)") K
                    CALL EXIT(1)
                 ENDIF
              READ(5,"(2E16.7)") (X1(I),X2(I),I=1,N)
C
C.....CALCULATE SAMPLE MEAN, COVARIANCE MATRIX, AND MAX-MIN VALUES
C
              XE1(K)=XE2(K)=C11=C12=C22=0.
              XMX1=XMN1=X1(1)
              XMX2=XMN2=X2(1)
              DO DL01 I=1,N
              XE1(K) = XE1(K) + X1(I)
```

```
            XE2(K) = XE2(K) + X2(I)
            C11 = C11 + (X1(I)*X1(I))
            C22 = C22 + (X2(I)*X2(I))
            C12 = C12 + (X1(I)*X2(I))
            IF (X1(I).GT.XMX1)  XMX1=X1(I)
            IF (X1(I).LT.XMN1)  XMN1=X1(I)
            IF (X2(I).GT.XMX2)  XMX2=X2(I)
            IF (X2(I).LT.XMN2)  XMN2=X2(I)
DL01        CONTINUE
C
            C11 = (C11 - (XE1(K)*XE1(K)/N))/(N-1)
            C22 = (C22 - (XE2(K)*XE2(K)/N))/(N-1)
            C12 = (C12 - (XE1(K)*XE2(K)/N))/(N-1)
            XE1(K) = XE1(K)/N
            XE2(K) = XE2(K)/N
          ENDIF
        IF (C11.LE.0.)  THEN
            WRITE(59,"(/""ERROR--X1 VARIANCE NEG OR 0; SET #"",I5)") K
            CALL EXIT(1)
          ENDIF
        IF (C22.LE.0.)  THEN
            WRITE(59,"(/""ERROR--X2 VARIANCE NEG OR 0; SET #"",I5)") K
            CALL EXIT(1)
          ENDIF
        S1(K) = 1./SQRT(C11)
        S2(K) = 1./SQRT(C22)
        CORCOEF = S1*S2*C12
C
C.....DIAGONALIZE (ROTATE) THE MATRIX
C
        IF (C12.EQ.0.)  THEN
          THETA = 0.
          ELSE
          IF (C11.EQ.C22)  THEN
            IF (C12.LT.0.)  THEN
                THETA = -PI/4.
              ELSE
                THETA = PI/4.
            ENDIF
            ELSE
            THETA = .5*ATAN(2.*C12/(C11-C22))
            IF (C11.LT.C22)  THEN
              IF (THETA.LT.0.)  THEN
                THETA=THETA+PI/2.
                ELSE
                THETA = THETA - PI/2.
              ENDIF
            ENDIF
          ENDIF
        ENDIF
C
        DISCR = SQRT((C11-C22)**2 + 4.*C12*C12)
        D1 = (C11+C22+DISCR)/2.
        D2 = (C11+C22-DISCR)/2.
        IF (D2.LT.0.)  D2=0.
C
C.....COMPUTE A CONSTANTS
C
        D1S = SQRT(D1)
        D2S = SQRT(D2)
```

```
      THC = COS(THETA)
      THS = SIN(THETA)
      A11 = D1S*THC
      A12 = -D2S*THS
      A21 = D1S*THS
      A22 = D2S*THC
C
C.....WRITE THE INFORMATION INTO THE PLOT FILE
C
      WRITE(100,"(//,""DATA SET NO. "",I2)") K
      WRITE(100,"(22X,""SAMPLE MEAN OF X1 : "",E12.3)") XE1(K)
      WRITE(100,"(22X,""SAMPLE MEAN OF X2 : "",E12.3)") XE2(K)
      WRITE(100,"(18X,""SAMPLE VARIANCE OF X1 : "",E12.3)") C11
      WRITE(100,"(18X,""SAMPLE VARIANCE OF X2 : "",E12.3)") C22
      WRITE(100,"(22X,""SAMPLE COVARIANCE : "",E12.3)") C12
      WRITE(100,"(16X,""CORRELATION COEFFICIENT : "",E12.3)") CORCOEF
      THETA = 180.*THETA/PI
      WRITE(100,F10) THETA
F10   FORMAT("  ANGLE OF MAJOR AXIS FROM X1 (DEGREES) :  ",F8.4)
      WRITE(100,"(17X,""DIAGONALIZED VARIANCES : "",2E12.3)") D1,D2
C
C.....COMPUTE ELLIPSE CURVES
C
      DO DL04 J=1,L
      ARG = -2.*ALOG(1.-PR(J))
      B = SQRT(ARG)
C
      DO DL03 I=1,NP
      X(I,J,K) = A11*B*WX(I) + A12*B*WY(I) + XE1(K)
      Y(I,J,K) = A21*B*WX(I) + A22*B*WY(I) + XE2(K)
      IF (X(I,J,K).GT.XMX1)  XMX1=X(I,J,K)
      IF (X(I,J,K).LT.XMN1)  XMN1=X(I,J,K)
      IF (Y(I,J,K).GT.XMX2)  XMX2=Y(I,J,K)
      IF (Y(I,J,K).LT.XMN2)  XMN2=Y(I,J,K)
DL03  CONTINUE
DL04  CONTINUE
DL10  CONTINUE
C
C*************************************************************
C*    PLOT THE RESULTS
C*************************************************************
C
      CALL FRAME
      IF (ITFLAG.NE.0)  THEN
           IF (X1T.GT.XMX1)  XMX1=X1T
           IF (X1T.LT.XMN1)  XMN1=X1T
           IF (X2T.GT.XMX2)  XMX2=X2T
           IF (X2T.LT.XMN2)  XMN2=X2T
        ENDIF
      AMX = .05*(XMX1-XMN1)
      AMY = .05*(XMX2-XMN2)
      XMX1 = XMX1 + AMX
      XMN1 = XMN1 - AMX
      XMX2 = XMX2 + AMY
      XMN2 = XMN2 - AMY
C
      CALL MAPS(XMN1,XMX1,XMN2,XMX2,.1,.999,.15,.999)
      IF (ITFLAG.NE.0)  CALL POINTC(1H*,X1T,X2T,1)
C
      NCHAR = 1HA
```

```
      DO DL06 K=1,M
      CALL POINTS(X1(1,K),X2(1,K),N)
      CALL LINEP(XE1(K),XMN2,XE1(K),XMX2,4)
      CALL LINEP(XMN1,XE2(K),XMX1,XE2(K),4)
C
      DO DL05 J=1,L
      CALL SETCRT(X(1,J,K),Y(1,J,K),0,1777B)
      CALL TRACEC(NCHAR,X(1,J,K),Y(1,J,K),NP)
DL05  CONTINUE
      NCHAR=NCHAR+1000000000000000000B
DL06  CONTINUE
C
      CALL SETCH(42.,3.,1,0,1,0)
      WRITE(100,F0) LBLX
      CALL SETCH(1.,42.,1,0,1,1)
      WRITE(100,F0) LBLY
F0    FORMAT(A10)
      CALL FRAME
C
C***************************************************************
C*    COMPUTE THE NORMALIZED RESULTS
C***************************************************************
C
      XMN1=XMX1=XMN2=XMX2=0.
C
      DO DL22 K=1,M
      DO DL21 J=1,L
      DO DL20 I=1,NP
      X(I,J,K) = S1(K)*(X(I,J,K)-XE1(K))
      Y(I,J,K) = S2(K)*(Y(I,J,K)-XE2(K))
      IF (X(I,J,K).GT.XMX1)  XMX1=X(I,J,K)
      IF (X(I,J,K).LT.XMN1)  XMN1=X(I,J,K)
      IF (Y(I,J,K).GT.XMX2)  XMX2=Y(I,J,K)
      IF (Y(I,J,K).LT.XMN2)  XMN2=Y(I,J,K)
DL20  CONTINUE
DL21  CONTINUE
DL22  CONTINUE
C
C***************************************************************
C*    PLOT THE NORMALIZED RESULTS
C***************************************************************
C
      AMX = .05*(XMX1-XMN1)
      AMY = .05*(XMX2-XMN2)
      XMX1 = XMX1 + AMX
      XMN1 = XMN1 - AMX
      XMX2 = XMX2 + AMY
      XMN2 = XMN2 - AMY
C
      CALL MAPS(XMN1,XMX1,XMN2,XMX2,.1,.999,.15,.999)
      CALL LINEP(0.,XMN2,0.,XMX2,4)
      CALL LINEP(XMN1,0.,XMX1,0.,4)
C
      NCHAR = 1HA
      DO DL25 K=1,M
C
      DO DL24 J=1,L
      CALL SETCRT(X(1,J,K),Y(1,J,K),0,1777B)
      CALL TRACEC(NCHAR,X(1,J,K),Y(1,J,K),NP)
DL24  CONTINUE
```

```
      NCHAR=NCHAR+1000000000000000000B
DL25  CONTINUE
C
      CALL SETCH(42.,3.,1,0,1,0)
      WRITE(100,F1) LBLX
      CALL SETCH(1.,42.,1,0,1,1)
      WRITE(100,F1) LBLY
F1    FORMAT(A10," - NORMALIZED")
C
C**************************************************************
C*     END PLOTTING -- CHECK THE TIME, THEN QUIT
C**************************************************************
C
      CALL TICHEK(IT1,IT2)
      IT2=IT2/1000000
      WRITE(59,FM1) IT2,IT1
FM1   FORMAT(/,I6,A10)
C
      CALL EXIT(1)
      END
```

APPENDIX B

Proof of Equations 6 and 7

Problem: Transform a known 2 x 2 symmetric matrix ($\underline{C}$) to a diagonalized form using an orthogonal transformation; i.e., find d_1^2, d_2^2, and θ from the equations:

$$\underline{C} = \underline{SDS}^{-1}$$

where $$\underline{D} = \begin{bmatrix} d_1^2 & 0 \\ 0 & d_2^2 \end{bmatrix}$$

$$\underline{S} = \begin{bmatrix} \cos\theta & -\sin\theta \\ \sin\theta & \cos\theta \end{bmatrix}$$

$$\underline{C} = \begin{bmatrix} c_{11} & c_{12} \\ c_{12} & c_{22} \end{bmatrix}$$

By simply multiplying through the right side,

$$c_{11} = d_1^2 \cos^2\theta + d_2^2 \sin^2\theta \tag{B-1}$$

$$c_{22} = d_1^2 \sin^2\theta + d_2^2 \cos^2\theta \tag{B-2}$$

$$c_{12} = \sin\theta \cos\theta \, (d_1^2 - d_2^2) \tag{B-3}$$

Now by subtracting (B-2) from (B-1):

$$c_{11} - c_{22} = (d_1^2 - d_2^2)\ (\cos^2\theta - \sin^2\theta)$$

$$= (d_1^2 - d_2^2) \cos 2\theta \tag{B-4}$$

then (B-3) becomes:

$$c_{12} = \sin\theta \cos\theta \left[\frac{c_{11} - c_{22}}{\cos 2\theta}\right]$$

$$= \left(\frac{\sin 2\theta}{2}\right)\left(\frac{1}{\cos 2\theta}\right)(c_{11} - c_{22})$$

which gives

$$\tan 2\theta = \frac{2c_{12}}{c_{11} - c_{22}}$$

and

$$\theta = 1/2 \tan^{-1} \frac{2c_{12}}{c_{11} - c_{22}} \tag{B-5}$$

Now, summing equation (B-1) and (B-2), we get

$$c_{11} + c_{22} = d_1^2 + d_2^2 \tag{B-6}$$

So, with (B-4) and (B-6)

$$\left.\begin{matrix} d_1^2 \\ \\ d_2^2 \end{matrix}\right\} = \frac{(c_{11} + c_{22}) \pm \dfrac{c_{11} - c_{22}}{\cos 2\theta}}{2} \tag{B-7}$$

But, from (B-5)

$$\cos 2\theta = \cos\left[\tan^{-1}\left(\frac{2c_{12}}{c_{11} - c_{22}}\right)\right]$$

$$= \frac{c_{11} - c_{22}}{\sqrt{(c_{11} - c_{22})^2 + 4\,(c_{12})^2}}$$

so,

$$\left.\begin{array}{l} d_{11}^2 \\ \\ d_{22}^2 \end{array}\right\} = \frac{c_{11} + c_{22} \pm \sqrt{(c_{11} - c_{22})^2 + 4c_{12}^2}}{2} \qquad \text{(B-8)}$$

Equations (B-5) and (B-8) are equivalent to equations (6) and (7). This can also be proved by noting that d_1^2 and d_2^2 are merely the eigenvalues of $\underline{C}$ and that the columns of $\underline{S}$ are the eigenvectors. Then, by solution of the eigenvalue-eigenvector problem, we achieve the same results.

When evaluating these equations by computer, we must be certain that the eigenvalues correctly correspond to their associate eigenvector. Otherwise, a 90° shift of the ellipse may result.